Shader "Unlit/NewUnlitShader"
{
    Properties
    {
        _MainTex ("Texture", 2D) = "white" {}
    }
    SubShader
    {
        Tags { "RenderType"="Opaque" }
        LOD 100

        Pass
        {
            CGPROGRAM
            #pragma vertex vert
            #pragma fragment frag
            #pragma multi_compile_fog
            // 支持GPU Instancing，材质面版会出现 Enable Instancing 复选框
            // 这句代码会生成两个变体，一个具有GPU实例化支持，一个不具有GPU实例化支持
            #pragma multi_compile_instancing

            #include "UnityCG.cginc"

            // 定义GPU Instancing的参数，名称为 Props，参数有2个
            UNITY_INSTANCING_BUFFER_START(Props)
                UNITY_DEFINE_INSTANCED_PROP(float4,_Color)
	      	    UNITY_DEFINE_INSTANCED_PROP(float, _Phi)
            UNITY_INSTANCING_BUFFER_END(Props)
            
            struct appdata
            {
                float4 vertex : POSITION;
                float2 uv : TEXCOORD0;

                // instanceID 加入顶点着色器输入结构
                UNITY_VERTEX_INPUT_INSTANCE_ID
            };

            struct v2f
            {
                float2 uv : TEXCOORD0;
                UNITY_FOG_COORDS(1)
                float4 vertex : SV_POSITION;
                
                // instanceID 加入顶点着色器输出结构
                UNITY_VERTEX_INPUT_INSTANCE_ID
            };

            sampler2D _MainTex;
            float4 _MainTex_ST;

            v2f vert (appdata v)
            {
                v2f o;

                // instanceID 在顶点的相关设置  
                UNITY_SETUP_INSTANCE_ID(v);
                // 传递 instanceID 顶点到片元
                UNITY_TRANSFER_INSTANCE_ID(v, o);

                // 获取传入的GPU Instancing参数
                float phi = UNITY_ACCESS_INSTANCED_PROP(Props, _Phi);
                v.vertex = v.vertex + sin(_Time.y + phi);
                
                o.vertex = UnityObjectToClipPos(v.vertex);
                o.uv = TRANSFORM_TEX(v.uv, _MainTex);
                UNITY_TRANSFER_FOG(o,o.vertex);
                return o;
            }

            fixed4 frag (v2f i) : SV_Target
            {
                // instanceID 在片元的相关设置
                UNITY_SETUP_INSTANCE_ID(i);
                
                fixed4 col = tex2D(_MainTex, i.uv);
                UNITY_APPLY_FOG(i.fogCoord, col);

                // 获取传入的GPU Instancing参数
                float4 ins_col= UNITY_ACCESS_INSTANCED_PROP(Props, _Color);
                
                return col * ins_col;
            }
            ENDCG
        }
    }
}
